



Vault Guardians Audit Report

Version 1.0

Cyfrin.io

May 17, 2024

Vault Guardians Audit Report

Bluedragon101

May 17, 2024

Prepared by: Bluedragon101

Lead Auditors: - [Shibi_Kishore]

Table of Contents

- Table of Contents
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
- Protocol Summary
 - Roles
- Executive Summary
 - Issues Found
 - High
 - * [H-1] Lack of UniswapV2 slippage protection in `UniswapAdapter::_uniswapInvest` enables frontrunners to steal profits
 - * [H-2] `ERC4626::totalAssets` checks the balance of vault's underlying asset even when the asset is invested, resulting in incorrect values being returned
 - * [H-3] Guardians can infinitely mint `VaultGuardianTokens` and take over DAO, stealing DAO fees and maliciously setting parameters
 - Medium

- * [M-1] Potentially incorrect voting period and delay in governor may affect governance
- Low
 - * [L-1] Incorrect vault name and symbol
 - * [L-2] Unassigned return value when divesting AAVE funds

Disclaimer

The Bluedragon101 team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following commit hash:

```
1 XXXX
```

Scope

```
1 ./src/
2 #-- abstract
3 |   #-- AStaticTokenData.sol
4 |   #-- AStaticUSDCData.sol
5 |   #-- AStaticWethData.sol
6 #-- dao
7 |   #-- VaultGuardianGovernor.sol
8 |   #-- VaultGuardianToken.sol
9 #-- interfaces
10 |  #-- IVaultData.sol
11 |  #-- IVaultGuardians.sol
12 |  #-- IVaultShares.sol
13 |  #-- InvestableUniverseAdapter.sol
14 #-- protocol
15 |  #-- VaultGuardians.sol
16 |  #-- VaultGuardiansBase.sol
17 |  #-- VaultShares.sol
18 |  #-- investableUniverseAdapters
19 |      #-- AaveAdapter.sol
20 |      #-- UniswapAdapter.sol
21 #-- vendor
22 |  #-- DataTypes.sol
23 |  #-- IPool.sol
24 |  #-- IUniswapV2Factory.sol
25 |  #-- IUniswapV2Router01.sol
```

Protocol Summary

This protocol allows users to deposit certain ERC20s into an ERC4626 vault managed by a human being, or a `vaultGuardian`. The goal of a `vaultGuardian` is to manage the vault in a way that maximizes the value of the vault for the users who have deposited money into the vault.

Roles

There are 4 main roles associated with the system.

- *Vault Guardian DAO*: The org that takes a cut of all profits, controlled by the `VaultGuardianToken`. The DAO that controls a few variables of the protocol, including:
 - `s_guardianStakePrice`
 - `s_guardianAndDaoCut`
 - And takes a cut of the ERC20s made from the protocol

- *DAO Participants*: Holders of the `VaultGuardianToken` who vote and take profits on the protocol
- *Vault Guardians*: Strategists/hedge fund managers who have the ability to move assets in and out of the investable universe. They take a cut of revenue from the protocol.
- *Investors*: The users of the protocol. They deposit assets to gain yield from the investments of the Vault Guardians.

Executive Summary

The Vault Guardians project takes novel approaches to work ERC-4626 into a hedge fund of sorts, but makes some large mistakes on tracking balances and profits.

Issues Found

Severity	Number of issues found
High	3
Medium	1
Low	2
Info	0
Gas	0
Total	6

High

[H-1] Lack of UniswapV2 slippage protection in `UniswapAdapter::_uniswapInvest` enables frontrunners to steal profits

Description: In `UniswapAdapter::_uniswapInvest` the protocol swaps half of an ERC20 token so that they can invest in both sides of a Uniswap pool. It calls the `swapExactTokensForTokens` function of the `UniswapV2Router01` contract, which has two input parameters to note:

```
1     function swapExactTokensForTokens(  
2         uint256 amountIn,  
3     @>    uint256 amountOutMin,  
4         address[] calldata path,
```

```
5         address to,  
6 @>      uint256 deadline  
7     )
```

The parameter `amountOutMin` represents how much of the minimum number of tokens it expects to return. The `deadline` parameter represents when the transaction should expire.

As seen below, the `UniswapAdapter::_uniswapInvest` function sets those parameters to 0 and `block.timestamp`:

```
1     uint256[] memory amounts = i_uniswapRouter.swapExactTokensForTokens  
2         (  
3 @>     amountOfTokenToSwap,  
4         0,  
5         s_pathArray,  
6 @>     address(this),  
7         block.timestamp  
8     );
```

Impact: This results in either of the following happening: - Anyone (e.g., a frontrunning bot) sees this transaction in the mempool, pulls a flashloan and swaps on Uniswap to tank the price before the swap happens, resulting in the protocol executing the swap at an unfavorable rate. - Due to the lack of a deadline, the node who gets this transaction could hold the transaction until they are able to profit from the guaranteed swap.

Proof of Concept:

1. User calls `VaultShares::deposit` with a vault that has a Uniswap allocation.
 1. This calls `_uniswapInvest` for a user to invest into Uniswap, and calls the router's `swapExactTokensForTokens` function.
2. In the mempool, a malicious user could:
 1. Hold onto this transaction which makes the Uniswap swap
 2. Take a flashloan out
 3. Make a major swap on Uniswap, greatly changing the price of the assets
 4. Execute the transaction that was being held, giving the protocol as little funds back as possible due to the `amountOutMin` value set to 0.

This could potentially allow malicious MEV users and frontrunners to drain balances.

Recommended Mitigation:

For the deadline issue, we recommend the following:

DeFi is a large landscape. For protocols that have sensitive investing parameters, add a custom parameter to the `deposit` function so the Vault Guardians protocol can account for the customizations of DeFi projects that it integrates with.

In the `deposit` function, consider allowing for custom data.

```
1 - function deposit(uint256 assets, address receiver) public override(  
    ERC4626, IERC4626) isActive returns (uint256) {  
2 + function deposit(uint256 assets, address receiver, bytes customData)  
    public override(ERC4626, IERC4626) isActive returns (uint256) {
```

This way, you could add a `deadline` to the Uniswap swap, and also allow for more DeFi custom integrations.

For the `amountOutMin` issue, we recommend one of the following:

1. Do a price check on something like a Chainlink price feed before making the swap, reverting if the rate is too unfavorable.
2. Only deposit 1 side of a Uniswap pool for liquidity. Don't make the swap at all. If a pool doesn't exist or has too low liquidity for a pair of ERC20s, don't allow investment in that pool.

Note that these recommendation require significant changes to the codebase.

[H-2] ERC4626::totalAssets checks the balance of vault's underlying asset even when the asset is invested, resulting in incorrect values being returned

Description: The `ERC4626::totalAssets` function checks the balance of the underlying asset for the vault using the `balanceOf` function.

```
1 function totalAssets() public view virtual returns (uint256) {  
2     return _asset.balanceOf(address(this));  
3 }
```

However, the assets are invested in the investable universe (Aave and Uniswap) which means this will never return the correct value of assets in the vault.

Impact: This breaks many functions of the `ERC4626` contract: - `totalAssets` - `convertToShares` - `convertToAssets` - `previewWithdraw` - `withdraw` - `deposit`

All calculations that depend on the number of assets in the protocol would be flawed, severely disrupting the protocol functionality.

Proof of Concept:

Code

Add the following code to the `VaultSharesTest.t.sol` file.

```
1 function testWrongBalance() public {
2     // Mint 100 ETH
3     weth.mint(mintAmount, guardian);
4     vm.startPrank(guardian);
5     weth.approve(address(vaultGuardians), mintAmount);
6     address wethVault = vaultGuardians.becomeGuardian(allocationData);
7     wethVaultShares = VaultShares(wethVault);
8     vm.stopPrank();
9
10    // prints 3.75 ETH
11    console.log(wethVaultShares.totalAssets());
12
13    // Mint another 100 ETH
14    weth.mint(mintAmount, user);
15    vm.startPrank(user);
16    weth.approve(address(wethVaultShares), mintAmount);
17    wethVaultShares.deposit(mintAmount, user);
18    vm.stopPrank();
19
20    // prints 41.25 ETH
21    console.log(wethVaultShares.totalAssets());
22 }
```

Recommended Mitigation: Do not use the OpenZeppelin implementation of the [ERC4626](#) contract. Instead, natively keep track of users total amounts sent to each protocol. Potentially have an automation tool or some incentivised mechanism to keep track of protocol's profits and losses, and take snapshots of the investable universe.

This would take a considerable re-write of the protocol.

[H-3] Guardians can infinitely mint VaultGuardianTokens and take over DAO, stealing DAO fees and maliciously setting parameters

Description: Becoming a guardian comes with the perk of getting minted Vault Guardian Tokens (vgTokens). Whenever a guardian successfully calls `VaultGuardiansBase::becomeGuardian` or `VaultGuardiansBase::becomeTokenGuardian`, `_becomeTokenGuardian` is executed, which mints the caller `i_vgToken`.

```
1 function _becomeTokenGuardian(IERC20 token, VaultShares tokenVault)
2     private returns (address) {
3     s_guardians[msg.sender][token] = IVaultShares(address(
4         tokenVault));
5     @> i_vgToken.mint(msg.sender, s_guardianStakePrice);
6     emit GuardianAdded(msg.sender, token);
7     token.safeTransferFrom(msg.sender, address(this),
8         s_guardianStakePrice);
```



```
6         token.approve(address(tokenVault), s_guardianStakePrice);
7         tokenVault.deposit(s_guardianStakePrice, msg.sender);
8         return address(tokenVault);
9     }
```

Guardians are also free to quit their role at any time, calling the `VaultGuardianBase::quitGuardian` function. The combination of minting `vgTokens`, and freely being able to quit, results in users being able to farm `vgTokens` at any time.

Impact: Assuming the token has no monetary value, the malicious guardian could accumulate tokens until they can overtake the DAO. Then, they could execute any of these functions of the `VaultGuardians` contract:

```
1     "sweepErc20s(address)": "942d0ff9",
2     "transferOwnership(address)": "f2fde38b",
3     "updateGuardianAndDaoCut(uint256)": "9e8f72a4",
4     "updateGuardianStakePrice(uint256)": "d16fe105",
```

Proof of Concept:

1. User becomes WETH guardian and is minted `vgTokens`.
2. User quits, is given back original WETH allocation.
3. User becomes WETH guardian with the same initial allocation.
4. Repeat to keep minting `vgTokens` indefinitely.

Code

Place the following code into `VaultGuardiansBaseTest.t.sol`

```
1     function testDaoTakeover() public hasGuardian hasTokenGuardian {
2         address maliciousGuardian = makeAddr("maliciousGuardian");
3         uint256 startingVoterUsdcBalance = usdc.balanceOf(
4             maliciousGuardian);
5         uint256 startingVoterWethBalance = weth.balanceOf(
6             maliciousGuardian);
7         assertEq(startingVoterUsdcBalance, 0);
8         assertEq(startingVoterWethBalance, 0);
9
10        VaultGuardianGovernor governor = VaultGuardianGovernor(payable(
11            vaultGuardians.owner()));
12        VaultGuardianToken vgToken = VaultGuardianToken(address(
13            governor.token()));
14
15        // Flash loan the tokens, or just buy a bunch for 1 block
16        weth.mint(mintAmount, maliciousGuardian); // The same amount as
17            the other guardians
18        uint256 startingMaliciousVGTokenBalance = vgToken.balanceOf(
19            maliciousGuardian);
```

```
14     uint256 startingRegularVGTokenBalance = vgToken.balanceOf(
15         guardian);
16     console.log("Malicious vgToken Balance:\t",
17         startingMaliciousVGTokenBalance);
18     console.log("Regular vgToken Balance:\t",
19         startingRegularVGTokenBalance);
20
21     // Malicious Guardian farms tokens
22     vm.startPrank(maliciousGuardian);
23     weth.approve(address(vaultGuardians), type(uint256).max);
24     for (uint256 i; i < 10; i++) {
25         address maliciousWethSharesVault = vaultGuardians.
26             becomeGuardian(allocationData);
27         IERC20(maliciousWethSharesVault).approve(
28             address(vaultGuardians),
29             IERC20(maliciousWethSharesVault).balanceOf(
30                 maliciousGuardian)
31         );
32         vaultGuardians.quitGuardian();
33     }
34     vm.stopPrank();
35
36     uint256 endingMaliciousVGTokenBalance = vgToken.balanceOf(
37         maliciousGuardian);
38     uint256 endingRegularVGTokenBalance = vgToken.balanceOf(
39         guardian);
40     console.log("Malicious vgToken Balance:\t",
41         endingMaliciousVGTokenBalance);
42     console.log("Regular vgToken Balance:\t",
43         endingRegularVGTokenBalance);
44 }
```

Recommended Mitigation: There are a few options to fix this issue:

1. Mint vgTokens on a vesting schedule after a user becomes a guardian.
2. Burn vgTokens when a guardian quits.
3. Simply don't allocate vgTokens to guardians. Instead, mint the total supply on contract deployment.

Medium

[M-1] Potentially incorrect voting period and delay in governor may affect governance

The `VaultGuardianGovernor` contract, based on OpenZeppelin Contract's Governor, implements two functions to define the voting delay (`votingDelay`) and period (`votingPeriod`). The contract intends to define a voting delay of 1 day, and a voting period of 7 days. It does it by returning the

value 1 `days` from `votingDelay` and 7 `days` from `votingPeriod`. In Solidity these values are translated to number of seconds.

However, the `votingPeriod` and `votingDelay` functions, by default, are expected to return number of blocks. Not the number seconds. This means that the voting period and delay will be far off what the developers intended, which could potentially affect the intended governance mechanics.

Consider updating the functions as follows:

```
1 function votingDelay() public pure override returns (uint256) {
2 -   return 1 days;
3 +   return 7200; // 1 day
4 }
5
6 function votingPeriod() public pure override returns (uint256) {
7 -   return 7 days;
8 +   return 50400; // 1 week
9 }
```

Low

[L-1] Incorrect vault name and symbol

When new vaults are deployed in the `VaultGuardianBase::becomeTokenGuardian` function, symbol and vault name are set incorrectly when the `token` is equal to `i_tokenTwo`. Consider modifying the function as follows, to avoid errors in off-chain clients reading these values to identify vaults.

```
1 else if (address(token) == address(i_tokenTwo)) {
2   tokenVault =
3   new VaultShares(IVaultShares.ConstructorData({
4     asset: token,
5 -     vaultName: TOKEN_ONE_VAULT_NAME,
6 +     vaultName: TOKEN_TWO_VAULT_NAME,
7 -     vaultSymbol: TOKEN_ONE_VAULT_SYMBOL,
8 +     vaultSymbol: TOKEN_TWO_VAULT_SYMBOL,
9     guardian: msg.sender,
10    allocationData: allocationData,
11    aavePool: i_aavePool,
12    uniswapRouter: i_uniswapV2Router,
13    guardianAndDaoCut: s_guardianAndDaoCut,
14    vaultGuardian: address(this),
15    weth: address(i_weth),
16    usdc: address(i_tokenOne)
17  }));
```

Also, add a new test in the `VaultGuardiansBaseTest.t.sol` file to avoid reintroducing this error, similar to what's done in the test `testBecomeTokenGuardianTokenOneName`.

[L-2] Unassigned return value when divesting AAVE funds

The `AaveAdapter::_aaveDivest` function is intended to return the amount of assets returned by AAVE after calling its `withdraw` function. However, the code never assigns a value to the named return variable `amountOfAssetReturned`. As a result, it will always return zero.

While this return value is not being used anywhere in the code, it may cause problems in future changes. Therefore, update the `_aaveDivest` function as follows:

```
1 function _aaveDivest(IERC20 token, uint256 amount) internal returns (
2     uint256 amountOfAssetReturned) {
3     - i_aavePool.withdraw({
4     + amountOfAssetReturned = i_aavePool.withdraw({
5         asset: address(token),
6         amount: amount,
7         to: address(this)
8     });
9 }
```